enter

Lecture 2-4 { Fri, Sept. 20}
More examples of boundary value problems in magnetostatics

Section 5.11. Magnetized sphere in an external field

We' ll find the solution in a clever way. Start with a uniformly magnetized sphere; i.e., magnetization = $\vec{M}$ inside the sphere.

Now superimpose a uniform magnetic induction $\vec{B}_0 = \mu_0 \vec{H}_0$ throughout all space. (*Question: Why is the superposition a solution of the field equations?*)

Now, inside the sphere we have

$$\vec{B}_{in} = \vec{B}_0 + \frac{2}{3}\mu_0 \vec{M}$$

$$\vec{H}_{in} = \frac{1}{\mu_0}\vec{B}_0 - \frac{1}{3}\vec{M}$$

Sphere of linear magnetic media

If the sphere is diamagnetic or paramagnetic, then $\vec{B}_{\text{in}} = \mu\, \vec{H}_{\text{in}}$ where μ = the permeability of the material.

$$\therefore \vec{B}_0 + \frac{2}{3}\mu_0 \vec{M} = \mu\left(\frac{1}{\mu_0}\vec{B}_0 - \frac{1}{3}\vec{M}\right)$$

Solve for the magnetization,

$$\vec{M} = \frac{3}{\mu_0}\left(\frac{\mu - \mu_0}{\mu + 2\mu_0}\right)\vec{B}_0 \quad \text{[linear medium]}$$

This answers the question, “What is the magnetization of a sphere with permeability μ in an applied magnetic induction?”

A ferromagnetic sphere

We can’t use that result for a ferromagnetic material; the constitutive equation is not linear, and not even single-valued (hysteresis).

Ferromagnetic materials (iron, nickel, cobalt and various alloys; NdFeB) may be permanently magnetized. I.e., $\vec{M}$ is not zero when $\vec{B}_0 = 0$.

So the earlier equation for $\vec{M}$ versus $\vec{B}_0$ is obviously not valid for a ferromagnetic object.

So now what?

Ferromagnetic sphere in a magnetic field $\vec{B}_0$

Here is what we know

$$B_{in} = B_0 + \frac{2\,\mu_0}{3}\,M$$

$$H_{in} = \frac{1}{\mu_0}B_0 - \frac{1}{3}\,M$$

Therefore,

$$B_{in} + 2\,\mu_0\,H_{in} = 3\,B_0$$

And we know something else: the hysteresis curve ...



Fig. 5.13

$$\vec{B} = \vec{F}(\vec{H})$$

Knowing $\vec{F}(\vec{H})$ we can solve for $\vec{B}_{\rm in}$ and $\vec{H}_{\rm in}$; the line with slope -2 and y intercept $3B_0$; the point P corresponds to $B_0 = 0$ and $M = -3\,H$.

Magnetic Shielding by a high permeability material

Section 5.12 : A spherical shell of permeable material in a magnetic field

fig515



Fig. 5.15 Shielding effect of a shell of highly permeable material.

You can read the calculation in Jackson. Here, instead, I'll do the calculation by Mathematica.


```
In[•]:= Remove["Global`*"]
(* PARAMETERS *)
(* a = inner radius; b = outer radius *)
(* μ' = the relative permeability *)
(* no free currents ⇒ H = -grad Φ_M *)
(* B = μ H ⇒ ∇^2 Φ_M = 0 *)
(* GUESS THE SOLUTIONS *)
Φext[r_, θ_] =
  -H0 * r * Cos[θ] + α1 / r^2 * Cos[θ];
Φshell[r_, θ_] =
  β1 * r * Cos[θ] + γ1 / r^2 * Cos[θ];
Φint[r_, θ_] = δ1 * r * Cos[θ];
```

```
In[•]:= (* BOUNDARY CONDITIONS *)
δHtb = D[Φext[r, θ], θ] - D[Φshell[r, θ], θ] /. {r -> b};
δHtb = δHtb / Sin[θ] // Simplify;
δHta = D[Φshell[r, θ], θ] - D[Φint[r, θ], θ] /. {r -> a};
δHta = δHta / Sin[θ] // Simplify;
δDnb =
  D[Φext[r, θ], r] - μ' * D[Φshell[r, θ], r] /. {r -> b};
δDnb = δDnb / Cos[θ] // Simplify;
δDna =
  μ' * D[Φshell[r, θ], r] - D[Φint[r, θ], r] /. {r -> a};
δDna = δDna / Cos[θ] // Simplify;
Eqs = {δHtb == 0, δHta == 0, δDnb == 0, δDna == 0};
Eqs // TableForm
```

$$\frac{-\alpha 1+b^3\,(H0+\beta 1)+\gamma 1}{b^2} == 0$$

$$-\frac{\gamma 1}{a^2}+a\,(-\beta 1+\delta 1) == 0$$

$$-H0-\frac{2\,\alpha 1}{b^3}+\left(-\beta 1+\frac{2\,\gamma 1}{b^3}\right)\mu' == 0$$

$$-\delta 1+\left(\beta 1-\frac{2\,\gamma 1}{a^3}\right)\mu' == 0$$

These are the boundary conditions; Eq. 5.120.


```
In[•]:= solution = Solve[Eqs, {α1 , β1, γ1, δ1}];
Length[solution];
αS = α1 /. solution[[1]] // Simplify;
βS = β1 /. solution[[1]] // Simplify;
γS = γ1 /. solution[[1]] // Simplify;
δS = δ1 /. solution[[1]] // Simplify;
In[•]:= FullSimplify[Numerator[δS]]
Normal[Series[Denominator[δS], {a, 0, 3}]]
```

$$-9\,b^3\,H0\,\mu'$$

$$2\,b^3+5\,b^3\,\mu'+2\,b^3\,(\mu')^2+a^3\left(-2+4\,\mu'-2\,(\mu')^2\right)$$

This is $\delta 1$ (the inner solution, numerator and denominator) ; equivalent to Eq. (5.121).

```
In[•]:= r = Sqrt[x^2 + z^2];
θ = ArcTan[z, x];
αC = αS /. {a → 1, b → 2, H0 → 1, μ' → 10};
BEx[x_, z_] = (-D[Φext[r, θ], x] /. {α1 → αC, H0 → 1}) * HeavisideTheta[r - 2];
BEz[x_, z_] = (-D[Φext[r, θ], z] /. {α1 → αC, H0 → 1}) * HeavisideTheta[r - 2];
pt1 = StreamPlot[{BEz[z, x], BEx[z, x]}, {x, -3, 3}, {z, -3, 3},
  StreamScale → {0.15, 0.15}, StreamStyle → Thickness[0.005]]
```




```
In[•]:= βC = βS /. {a → 1, b → 2, H0 → 1, μ' → 10};
γC = γS /. {a → 1, b → 2, H0 → 1, μ' → 10};
BSx[x_, z_] = 10 * (-D[Φshell[r, θ], x] /. {β1 → βC, γ1 → γC, H0 → 1}) *
   HeavisideTheta[2 - r] * HeavisideTheta[r - 1];
BSz[x_, z_] = 10 * (-D[Φshell[r, θ], z] /. {β1 → βC, γ1 → γC, H0 → 1}) *
   HeavisideTheta[2 - r] * HeavisideTheta[r - 1];
pt2 = StreamPlot[{BSz[z, x], BSx[z, x]}, {x, -3, 3}, {z, -3, 3},
  StreamScale → {0.15, 0.15}, StreamStyle → Thickness[0.005]]
```



```
In[•]:= δC = δS /. {a → 1, b → 2, H0 → 1, μ' → 10};
BIx[x_, z_] = (-D[Φint[r, θ], x] /. {δ1 → δC, H0 → 1}) * HeavisideTheta[1 - r];
BIz[x_, z_] = (-D[Φint[r, θ], z] /. {δ1 → δC, H0 → 1}) * HeavisideTheta[1 - r];
pt3 = StreamPlot[{BIz[z, x], BIx[z, x]}, {x, -3, 3}, {z, -3, 3},
  StreamScale -> {0.15, 0.15}, StreamStyle → Thickness[0.005]]
```




```
In[•]:= Show[pt1, pt2, pt3,
 Graphics[{Red, Thickness[0.003],
   Circle[{0, 0}, 1], Circle[{0, 0}, 2]}],
 FrameLabel → {"z", "x"},
 PlotLabel → "Magnetic shielding;
K = μ/μ0 = 10 , b = 2a",
 ImageSize → Large]
```



Shielding

Plot $B_z(0,0,z)$ versus z.   N.B.: B_{normal} is continuous.

```
In[•]:= Plot[
  {BEz[0, z], BSz[0, z], BIz[0, z]},
  {z, -5, 5}, PlotRange → {{-5, 5}, {0, 4}},
  PlotStyle → {{Thickness[0.006], Black}},
  AxesLabel → {"z", "Bz[z]"},
  Epilog → {Thickness[0.004],
    Dashing[0.02], Line[{{-5, 1}, {5, 1}}]},
  ImageSize → Large, BaseStyle → {24}]
```



Plot $B_z(x,0,0)$ versus x. N.B.: $B_{\text{tangential}}$ is not continuous.

```
In[•]:= Plot[
  {BEz[x, 0], BSz[x, 0], BIz[x, 0]},
  {x, -5, 5}, PlotRange → {{-5, 5}, {0, 4}},
  PlotStyle → {{Thickness[0.006], Black}},
  AxesLabel → {"x", "Bz[x]"},
  Epilog → {Dashing[0.02], Line[{{-5, 1}, {5, 1}}]},
  ImageSize → Large, BaseStyle → {24}]
```



The shielding occurs because the high permeability shell sucks in the magnetic field lines.



Fig. 5.15 Shielding effect of a shell of highly permeable material.